Aula prática n.7 – Variáveis Instrumentais

Como nas aulas anteriores, abra o Stata e configure-o por meio do arquivo ‘profile.do’.

Para sua comodidade, a configuração abaixo está incluída no arquivo ‘stata_aula7.do’, que se encontra no endereço

 http://www.undp-povertycentre.org/evaluation/praticas/do/. 

Se você preferir, use diretamente o ‘profile’ desse último arquivo.
clear

set mem 100m

cd "E:\curso"

sysdir set PLUS "E:\ado\plus"

sysdir set PERSONAL "E:\ado\personal"

sysdir set OLDPLACE "E:\ado"

1. A base de dados
A base de dados para esta aula chama-se ‘nls80.dta’ e encontra-se no link abaixo:

http://www.undp-povertycentre.org/evaluation/praticas/bases/
Faça o download, lembrando de NÃO salvá-los como .HTML.
Apesar de utilizarmos as informações relativas ao ano de 1980, a National Longitudinal Survey é, na realidade, um painel, contendo informações relacionadas principalmente ao mercado de trabalho.
Vamos abrir um arquivo .log e em seguida, abrir a base.

log using log\aula7.log

use base\nls80.dta, clear

Agora, vamos investigar a base.

describe
summarize
2. Educação como variável endógena
Como visto no exemplo em aula, é possível que a variável ‘anos de estudo’ seja endógena em uma equação salarial. Assim, esse será o nosso objeto de estudo. A variável ‘educ’ na base de dados encontra-se na forma contínua. Vamos criar uma variável dummy de tratamento, indicando se o indivíduo possui mais de 12 anos de estudo, ou seja, ingressou na faculdade. Vamos fazer o mesmo com a educação dos pais, para considerar um instrumento discreto. Vamos considerar uma variável indicando se os pais concluíram o segundo grau. A idéia desses instrumentos é simples: pais com mais educação são propensos a prover mais educação aos filhos. No entanto, podemos assumir que a educação dos pais não reflete na produtividade dos filhos.

g deduc = educ > 12

foreach name in feduc meduc {

    generate d`name' = `name'>11

}
Outro possível instrumento é a ordem de nascimento. Pode-se argumentar que, devido à restrição orçamentária da família (ou algum outro fator não observável), os pais priorizam a educação de um dos filhos. Vamos construir uma variável dummy indicando se o indivíduo é filho mais velho. O salário do indivíduo não é determinado por esse fato.
gen primeiro = brthord == 1

3. Estimando por OLS
Primeiramente, vamos estimar uma equação salarial considerando a educação como variável exógena. As variáveis ‘iq’ e ‘kww’ podem ser consideradas como proxies para habilidade. Vamos estimar dois modelos, excluindo e incluindo essas variáveis.

reg lwage deduc exper married age black south urban

reg lwage deduc iq kww exper married age black south urban

Podemos ver que as proxies são significativas. Vamos, portanto, mantê-las nos próximos modelos.
4. Estimando por Variável Instrumental
Agora, vamos estimar um LATE. Relembrando o que foi dito em aula, esse estimador nada mais a comparação entre as médias da variável de interesse e de tratamento, para dois distintos valores do instrumento. Podemos calcular o LATE da seguinte forma:
ivregress 2sls lwage (deduc =  dfeduc)

ivregress 2sls lwage (deduc =  primeiro)

Passemos para a estimação do ATE, em dois estágios. O comando utilizado já corrige a matriz de variância-covariância. Essa correção é necessária sempre que há uma estimação em dois estágios. Isso porque há uma perda de eficiência do estimador por conta da limitação do uso da informação disponível.

ivregress 2sls lwage iq kww exper married age black south urban (deduc =  dfeduc), small

ivregress 2sls lwage iq kww exper married age black south urban (deduc = primeiro), small

O estimador acima utiliza a projeção linear da variável endógena como regressor no segundo estágio. Vamos observar a diferença quando se utiliza a probabilidade predita como instrumento.
probit deduc age exper married black south urban dfeduc

predict edhat

ivregress 2sls lwage iq kww exper married age  black south urban (deduc = edhat), small
Note a diferença de significância entre os dois modelos anteriores. Wooldrigde (2002) afirma que para estimar o modelo com projeção linear é preciso que P(D=1|X,Z) esteja corretamente especificado. Ademais, a variância do estimador desse modelo depende da variância assintótica dos parâmetros do primeiro estágio.
Se você desconfia no comando ‘ivregress’, você pode estimar o ATE em dois estágios passo a passo. Nesse caso, é importante lembrar que os erros-padrões do segundo estágio não levam em conta o fato da estimação se realizada em dois estágios, e precisam ser corrigidos.
reg deduc iq kww exper married age black south urban dfeduc
predict edhatli

reg lwage edhatli iq kww exper married age black south urban, vce(bootstrap, rep(100))
Vamos incluir variáveis de interação na tentativa de melhorar o controle. Para fazer isso, inclui-se na regressão a diferença do valor observado do indivíduo em relação à média, para cada variável de controle, interagindo com a dummy de tratamento.
Criamos os desvios em relação à média, 

, e interagimos com a dummy de tratamento e a probabilidade estimada de participação do tratamento, esta última para utilizar como instrumento.

foreach name in iq kww exper tenure age married black south urban {

    qui sum `name'

    g dt`name' = deduc*(`name' - r(mean))

    g p`name' = ehat*(`name' - r(mean))

}
ivregress 2sls lwage iq kww exper married age black south urban dtiq-dturban (deduc = edhat piq-purban), small
Podemos também estimar o ATT, incluindo as interações.

foreach name in iq kww exper tenure age married black south urban {

    qui sum `name' if deduc==1

    g dtt`name' = deduc*(`name' - r(mean))

    g pt`name' = edhat*(`name' - r(mean))

}

ivregress 2sls lwage iq kww exper married age black south urban dttiq-dtturban (deduc = edhat ptiq-pturban), small
Assim como na aula 3, quando estimamos o impacto da presença de eletricidade na fecundidade em Botsuana, o ATE é maior que o ATT, de forma que podemos esperar que, quando os indivíduo que não entraram na universidade o fizerem, o retorno será ainda maior, o que representa um resultado não convencional, se pensarmos em termos de equilíbrio geral. Mas isso é outra história!

5. Propensity Score com Variável Instrumental

Conforme visto em aula, a variável de tratamento pode ser considerada uma restrição de exclusão na estimativa da média condicional da variável de interesse. Na presença de um intrumento, pode-se assumir que o tratamento é um caso especial quando todos os indivíduos são compliers, ou seja, quando Z = D. Assim, podemos estimar o ATE da seguinte forma:
reg deduc iq kww exper married age black south urban dfeduc

predict pxz

g pxz2 = pxz^2

reg pxz iq kww exper married age black south urban

predict px

reg pxz2 iq kww exper married age black south urban

predict qx

reg lwage pxz px qx
6. Endogenous switching model

Essa forma de estimação não foi abordada em aula, mas é uma alternativa à estimação por variável instrumental. Esse método estima a equação de interesse e a equação de tratamento conjuntamente, assumindo que os erros seguem uma distribuição normal bivariada. Assim, mesmo sem a presença de um instrumento, é possível estimar o modelo, embora possa haver problemas de convergência.

treatreg lwage iq kww exper married age black south urban, treat(deduc = dfeduc iq kww exper married age black south urban)

Exercícios

· A aula prática de hoje está no arquivo ‘stata_aula7.do’.

· Utilizando a base ‘mroz.dta’, disponível em

http://www.undp-povertycentre.org/evaluation/praticas/bases/,


procure identificar quais as variáveis de interesse e os possíveis instrumentos, justificando a escolha desses últimos. Compare as estimativas de LATE , ATE e ATT.

