Aula prática n.10 – Tratamento contínuo: Generalized Propensity Score
Imai & Dyk (2003)

O início do arquivo ‘stata_aula10.do’ contém as configurações básicas geralmente utilizadas pelo usuário de Stata:

clear

set mem 200m

cd "G:\curso"

sysdir set PLUS "E:\curso\ado\plus"

sysdir set PERSONAL "E:\curso\ado\personal"

sysdir set OLDPLACE "E:\curso\ado"
1. A base de dados
A base utilizada para esta aula prática do curso refere-se a uma pesquisa americana sobre gastos médicos “1987 National Medical Expenditure Survey” (NMES). A vantagem desta base é que ela inclui não só os custos médicos, mas também informação sobre a freqüência e a duração do vicio de fumar para um grande número de indivíduos.

A base inclui informação sobre: 

- Idade na época da pesquisa

- idade quando começou a fumar

- sexo

- raça

- estado civil

- nível educacional

- região

- pobreza

- uso do cinto de segurança

- gastos médicos (variável de interesse)
- maços por dia x números de ano que fumou (tratamento)
Definamos o arquivo de log e puxemos a base de dados que vamos utilizar:

log using log\aula9.log, replace

use http://www.undp-povertycentre.org/evaluation/praticas/bases/nmesdata.dta, clear
Selecionando, preparando e construindo as variáveis:

keep cigsaday yearsince packyears lastage male race3 beltuse agesmoke educate marital sregion povstalb totalexp hsqaccwt 

destring agesmoke agestop cigsaday, replace

rename race3 race

g lnpackyears = ln(packyears)

g ltotalexp = ln(totalexp)

g agesmoke2 = agesmoke^2

g lastage2 = lastage^2

g posexp = totalexp>0 

REGRESSAO SIMPLES

xi: reg ltotalexp lnpackyears lastage lastage2 male i.race i.beltuse agesmoke agesmoke2 i.educate i.marital i.sregion i.povstalb [aw=hsqaccwt]

PRIMEIRO PASSO: Modelando 

xi: reg lnpackyears lastage lastage2 male i.race i.beltuse agesmoke agesmoke2 i.educate i.marital i.sregion i.povstalb [aw=hsqaccwt]

SEGUNDO PASSO (A): Calculando 

predict index if e(sample)

SEGUNDO PASSO (B): TESTANDO A ESPECIFICACAO (Balanceamento)

foreach name in race beltuse educate marital sregion povstalb {


qui tab `name', g(`name')

}

mat result = ., .

foreach name in lastage lastage2 male agesmoke agesmoke2 race2 race3 beltuse2 beltuse3 educate2 educate3 educate4 marital2 marital3 marital4  marital5 sregion2 sregion3 sregion4 povstalb2 povstalb3 povstalb4 povstalb5 {


 
reg `name' lnpackyears [aw=hsqaccwt]



loc t`name' = _b[lnpackyears]/_se[lnpackyears]


 
reg `name' lnpackyears index [aw=hsqaccwt]



loc t`name'i = _b[lnpackyears]/_se[lnpackyears]



mat result = result \ `t`name'', `t`name'i'

}

mat list result

3o PASSO: subclassificar observações com valores similares de 
[image: image3.wmf]^

q

 em um numero parcimonioso de subclasses de tamanho parecidos.

Vamos utilizar o comando “quantiles”.
findit quantiles
quantiles index [aw= hsqaccwt] , gencatvar(class3) nquant(3)

quantiles index [aw= hsqaccwt] , gencatvar(class10) nquant(10)

4o PASSO: Dentro de cada subclasse modelar: 

e estimar o coeficiente da regressão de Y(tP) contra (tP) e calcular o efeito causal médio como a media ponderada do efeito de cada uma das subclasses

Com três subclasses:
tab class3 if lnpackyears ~=. & ltotalexp~=.

loc N3 = r(N)

di `N3'

loc beta = 0

forval n = 1/3 {


reg ltotalexp lnpackyears index if class3==`n'


loc b`n' = _b[lnpackyears]


loc se`n' = _se[lnpackyears]


loc n`n' = e(N)


di `n`n''


di `n`n''/`N3'


loc beta = `beta' + (`b`n'')*`n`n''/`N3'


di `beta'

}

Com dez subclasses:

tab class10 if lnpackyears ~=. & ltotalexp~=.

loc N10 = r(N)

di `N10'

loc beta = 0

forval n = 1/10 {


reg ltotalexp lnpackyears index if class10==`n'


loc b`n' = _b[lnpackyears]


loc se`n' = _se[lnpackyears]


loc n`n' = e(N)


di `n`n''


di `n`n''/`N10'


loc beta = `beta' + (`b`n'')*`n`n''/`N10'


di `beta'

}

log close

Exercícios

· Repita o exercício estimando um logit (tem despesa medica=1/0 – variável dependente) como função da variável lnpackyears.

