Aula prática n.3 – Estimação por meio de regressões
Abra o Stata e digite o comando:
do G:\curso\do\profile.do

Agora seu Stata está pronto para ser utilizado.
A aula prática de hoje está no arquivo ‘stata_aula3.do’.
Baixe o arquivo de http://www.undp-povertycentre.org/evaluation/praticas/do/
e o salve em “E:\curso\do” (lembrando de NÃO salvá-lo como HTML).

No Stata, digite:

doedit do\stata_aula3.do
Antes de tudo, crie um arquivo log:
log using log\aula3.log

1. A base de dados
A base utilizada para a terceira aula prática do curso refere-se de uma amostra de mulheres em Botsuana e serve para investigar efeitos sobre a fecundidade.

Para abrir o arquivo, basta digitar:
use http://www.undp-povertycentre.org/evaluation/praticas/bases/FERTIL2.dta, clear
Ou seja, quando uma base de dados está disponível on-line, o Stata é capaz de utilizá-la sem a necessidade de salvá-la no computador.

2. Análise descritiva
Para uma descrição do conteúdo da base de dados, utilize o comando describe:
describe
Para uma análise descritiva geral das variáveis, utilize o comando summarize:
summarize

Note que para algumas variáveis há um elevado número de informações ausentes. O motivo é que, por definição, nem todas as mulheres respondem estas perguntas (casos ‘não-aplicáveis’).
Nosso interesse é saber como a eletricidade influencia a fecundidade. A princípio, esta variável é exógena, pois a taxa de fecundidade não determina se uma região tem ou não energia elétrica.
Primeiro vejamos como se distribui a freqüência da variável de tratamento:

tabulate electric
Depois fazemos uma análise descritiva por sub-amostras de mulheres com e sem acesso à energia elétrica:

by electric, sort: summarize mnthborn yearborn age radio-evermarr
Repare que existe uma sub-amostra onde os valores são missing. Assumindo que a taxa de não-resposta é aleatória, estas observações são automaticamente omitidas da investigação.

Uma primeira sondagem a respeito da relação entre número de crianças e acesso à energia elétrica pode ser feita por meio de uma tabela:

tabulate children electric, row nofreq
3. Regressão Linear
Começaremos com uma regressão linear simples (univariada):
regress children electric
Para salvar o resultado desta regressão, digite:

estimates store OLSimple
Um impacto negativo é identificado, mas pode haver um problema de variáveis omitidas (correlação espúria). Vamos incluir mais variáveis no modelo e salvá-lo novamente:
regress children electric age agesq educ urban evermarr

estimates store OLSmult
Repare que o estimador de impacto mudou:

estimates table OLSimple OLSmult, star
Mas só a inclusão de variáveis de controle pode não bastar. Vamos gerar as variáveis de interação. Antes, precisamos saber qual a média da variável de controle:

summarize age if e(sample)
Utilizar a restrição if e(sample) é importante porque garante que a média obtida refere-se à amostra utilizada na regressão. e() é onde geralmente resultados de uma estimação (e-class) são salvos após um comando. r() é onde resultados em geral (r-class) são salvos após um comando.

Por exemplo, estamos interessados na média da variável age, que está salva, após o comando summarize, em r(mean). Vamos gerar a variável de interação:

generate M_age = electric*(age – r(mean))
Essa variável representa o quanto a idade das mulheres que têm acesso à eletricidade se distingue da idade média da amostra utilizada:

Pode ser um pouco trabalhoso gerar as variáveis de interação para cada atributo. No Stata, sempre quando nos defrontamos com rotinas extensas e repetitivas, apelamos para os comandos while, foreach, forvalues e, eventualmente, if (não confundir com o if de restrição sobre a amostra). A idéia destes comandos é repetir o mesmo procedimento diversas vezes, alterando somente uma parte pré-definida. No nosso caso, queremos estimar a média de uma variável e depois gerar uma nova variável a partir desta média:
foreach var of varlist agesq educ urban evermarr {
    summarize `var’ if e(sample)

    generate M_`var’ = electric*(`var’ – r(mean))

}

Importante: em um algoritmo sempre quando uma chave for aberta, é interessante utilizar indent. Isso nos identifica o nível do comando.
Agora, podemos estimar novamente a regressão, incluindo as variáveis de interação. Para não repetir todas as novas variáveis geradas no comando, vamos utilizar ‘*’:
regress children electric age agesq educ urban evermarr M_*
estimates store OLSate
Como esperado, a interação com a dummy de área urbana é significativa. Antes não sabíamos se a fecundidade era menor porque as mulheres viviam em áreas urbanas ou porque tinham acesso à luz elétrica. De fato, a dummy de área urbana explica (determina) positivamente a dummy de energia elétrica. Esta última perde seu poder de explicação quando a primeira é incluída na regressão.
Outro parâmetro de interesse que pode ser estimado é o ATT. Para isso basta centralizar o coeficiente de interesse na média das covariáveis do grupo de tratamento. . Para que o procedimento a ser realizado não apareça na janela de resultados do stata e aumente desnecessariamente o seu log file, utilizaremos o prefixo quietly:

quietly foreach var of varlist age agesq educ urban evermarr {
    summarize `var’ if e(sample) & electric==1
    generate S_`var’ = electric*(`var’ – r(mean))

}

Rodando a regressão:
regress children electric age agesq educ urban evermarr S_*
estimates store OLSatt
Comparando os resultados:
estimates table OLSimple OLSmult OLSate OLSatt, star
Note que o ATE é maior que o ATT. Ou seja, podemos esperar que o ATU seja ainda maior que o ATT. Portanto, se expandirmos a cobertura de luz elétrica em Botsuana, estimamos que a redução na fecundidade decorrente desta expansão será seja maior do que àquela ocorridau quando as atuais residências receberam luz elétrica.

4. Heterogeneidade do Impacto

Vamos voltar ao estimador ATE, para verificar se o impacto sobre quem tem menos de 7 anos de estudo é diferente do impacto sobre quem tem 7 anos de estudo ou mais.

Antes, precisamos saber o quanto cada sub-amostra se diferencia da média em termos de educação:
summarize M_educ if e(sample) & educ<7

Vamos salvar esta média em um escalar (ocupa menos espaço na memória do que gerar uma variável com este valor):

scalar Educ0 = r(mean)

Para o outro grupo:

summarize M_educ if e(sample) & educ>=7

scalar Educ7 = r(mean)

Agora, vamos rever o resultado da regressão:
estimates restore OLSate

Para testar a heterogeneidade, precisamos fazer um teste de combinação linear sobre coeficientes. Primeiro com o grupo com menor escolaridade:
lincom _b[electric] + _b[M_educ]*Educ0
Depois com o grupo de maior escolaridade:

lincom _b[electric] + _b[M_educ]*Educ7

Podemos ainda testar a diferença absoluta entre estes coeficientes:
lincom _b[M_educ]*Educ0 - _b[M_educ]*Educ7
E a diferença relativa (percentual) entre estes coeficientes, porém utilizando um comando para testar combinações não-lineares entre coeficientes:
nlcom (_b[M_educ]*Educ0 - _b[M_educ]*Educ7)/_b[electric]

5. Outros modelos de regressão

Talvez uma regressão OLS não seja a maneira mais consistente de se estimar o número de filhos. Uma sugestão é estimar um modelo de contagem, com distribuição Poisson:

poisson children electric age agesq educ urban evermarr M_*
O problema desses modelos é que o coeficiente pode não representar o parâmetro de interesse. Uma maneira fácil de calcular o efeito sobre a variação na média é com o comando mfx:

mfx, varlist(electric)

Outro modelo que pode ser utilizado neste caso é o probit (ou logit) ordenado, que estima a probabilidade de ter um filho a mais:

oprobit children electric age agesq educ urban evermarr M_*
Efeitos marginais:
* probabilidade de não ter filhos
mfx, varlist(electric) predict(outcome(0))

* probabilidade de ter um filho

mfx, varlist(electric) predict(outcome(1))

* probabilidade de ter mais um filho, dado que já tem 1
mfx, varlist(electric) predict(outcome(2))
* probabilidade de ter mais um filho, dado que já tem 2
mfx, varlist(electric) predict(outcome(3))
Ao invés de estimar o número de filhos, podemos estar interessados em verificar o efeito da luz elétrica sobre o desejo de ter mais ou menos filhos. Primeiramente, vamos gerar a variável que representa a diferença entre o número de filhos que a mulher considera ideal e o número de filhos que ela tem:
generate vontade = idlnchld – children

E confira a distribuição desta variável:
histogram vontade

Agora, vamos criar uma dummy que representa a vontade de ter mais filhos e uma dummy que representa a vontade de ter menos filhos:

generate maisfil = vontade>0 if vontade!=.

generate menosfil = vontade<0 if vontade!=.

Podemos estimar o impacto da luz elétrica sobre a probabilidade de querer ter mais filhos por meio de um modelo probit:

probit maisfil electric age agesq educ urban evermarr M_*

mfx, varlist(electric)
Da mesma forma, podemos estimar o impacto da luz elétrica sobre a probabilidade de querer ter menos filhos:

logit menosfil electric age agesq educ urban evermarr M_*
mfx, varlist(electric)
Por fim, feche o arquivo log, salve a base no seu pen-drive e feche o Stata:
log close
save base\FERTIL2.dta
exit
Exercício
· Utilizando a mesma base de dados, construa um modelo causal para ser estimado. Ou seja, defina um resultado de interesse e um parâmetro de impacto e depois estime o modelo que melhor responde a sua pergunta.
