Aula prática n.9 – Regressão Descontínua
A aula de hoje encontra-se na página abaixo, no arquivo ‘stata_aula9’:
 http://www.undp-povertycentre.org/evaluation/praticas/do/. 

Como de costume:

clear

set mem 100m

cd "E:\curso"

sysdir set PLUS "E:\ado\plus"

sysdir set PERSONAL "E:\ado\personal"

sysdir set OLDPLACE "E:\ado"

1. A base de dados
Nesta aula, utilizaremos três bases de dados. Todas estão na página abaixo.

http://www.undp-povertycentre.org/evaluation/praticas/bases/
A primeira chama-se ‘votex.dta’. A segunda chama-se ‘nordeste.dta’ e a terceira ‘nordestepess.dta’

Faça o download, lembrando de NÃO salvá-los como .HTML.
A primeira base apresenta informações sobre alguns distritos americanos, em especial dados sobre votações e a divisão democratas/republicanos. Utilizaremos essa base para trabalhar com o desenho Sharp.

A outras duas bases contêm informações para a região nordeste do Brasil, retirados da PNAD 2006. Enquanto a base ‘nordeste’ inclui os domicílios com renda per capita de até R$ 1000,00, a base ‘nordestepess’ traz dados sobre crianças entre 7 e 15 anos. Essas bases serão usadas para trabalhar com o desenho Fuzzy.
Vamos abrir um arquivo .log.
log using log\aula9.log

2. Sharp RDD
Vamos abrir a base ‘votex.dta’
use base\votex.dta, clear

Como visto em sala, o desenho Sharp ocorre quando o tratamento é determinístico. Nessa base, temos a informação de qual porcentagem dos eleitores votaram em candidatos democratas. Essa é a nossa variável de seleção. O tratamento é se o estado é democrata ou republicano. Estamos interessados no gasto federal naquele estado.

O que está contido na base.

summarize

Primeiramente, uma análise gráfica. Os comandos abaixo geram gráficos relacionando a variável de interesse, ‘lne’ (gastos), e a variável de seleção, ‘d’ (porcentagem de votação).
tw scatter lne d if d<=0 || scatter lne d if d>0, xline(0) 
tw lowess lne d if d<=0 || lowess lne d if d>0, xline(0)
Podemos ver que no ponto de corte há uma grande descontinuidade. O método de RDD basicamente avalia a diferença da variável de interesse na redondeza desse ponto de corte.

Agora, vamos observar o gráfico da variável de tratamento vs. variável de seleção.

tw scatter win d if d<=0 || scatter win d if d>0, xline(0)

Como temos apenas controles à esquerda do ponto corte, e apenas tratados à direita, o gráfico salda de zero pra um naquele ponto.
2.1 Regressão Não Paramétrica na Fronteira

Essa regressão nada mais é do que uma diferença de médias utilizando a informação relevante, determinada pela escolha do bandwidth. Vamos supor um bandwith de 0.1. Primeiro, faremos passo-a-passo, depois, usaremos um modo alternativo direto.
Passo-a-passo

g h = d>=-.1 & d<=.1

g yl = cond(h==1 & d<0, lne,0)

g yr = cond(h==1 & d>0, lne,0)

mean yl

loc xl = _b[yl]

mean yr

loc xr = _b[yr]

g obsl = h==1 & d<0

g obsr = h==1 & d>0

mean obsl

loc l = _b[obsl]

mean obsr

loc r = _b[obsr]

di (`xr'/`r') - (`xl'/`l')

Modo alternativo

mean lne if h==1, over(win)

lincom _b[1] - _b[0]

Ambos produzem o mesmo resultado. Isto é, apesar da descontinuidade observada graficamente (3%), não temos evidência de que essa diferença é significativa. Podemos alterar o bandwidth para analisar a sensibilidade do resultado. 
2.2 Regressão Linear Local
Vimos que o método de Kernel utilizado em regressões não paramétricas não se comporta bem em pontos situados na fronteira. A literatura sobre RDD recomenda utilizar esse método com regressão paramétrica, dando origem à regressão linear local. Para simplificar, continuamos com o Kernel retangular. Novamente, duas formas de estimação.

Passo-a-passo

reg lne d if h==1 & d<0

loc alfal = _b[_cons]

reg lne d if h==1 & d>0

loc alfar = _b[_cons]

di `alfar' - `alfal'

Alternativamente
g dwin = d*win

reg lne d win dwin if h==1

Podemos ver que o impacto é muito maior, mas continua não significativo.
O Stata possui um comando chamado rd.ado. Vamos baixá-lo da internet. Um meio de buscar comandos com nomes desconhecidos é utilizando o comando search.

search discontinuity, all

Esse comando estima o impacto por meio da regressão linear local. Para obter a significância do parâmetro, pode-se utilizar o bootstrap. 
rd lne d, gr mbw(100)

bootstrap, reps(10): rd lne d, gr mbw(100)

3. Fuzzy RDD
Agora, vamos estimar trabalhar com o desenho fuzzy. Para isso, vamos, inicialmente, utilizar a base de domicílios da pnad. Ela contém informações sobre renda domiciliar per capita e se algum membro recebe bolsa família. Queremos investigar o impacto do bolsa-família na freqüência à escola e no trabalho infantil.
use data\nordeste, clear

sum
Vamos recodificar a variável V2403, que indica o recebimento do bolsa-família, e gerar a variável de tratamento.

recode v2403 (3=0), copyrest g(treat)

Primeiramente, vamos observar se o tratamento realmente não é determinístico.

tab renda treat

Aqui segue uma dica do Stata. Queremos saber se as variáveis abaixo identificam unicamente o domicílio.

bys uf v0102 v0103: g d = _N

Agora, queremos saber a proporção de pessoas que recebem o bolsa-familia por valor de renda. Como nem todas as pessoas abaixo do valor de elegibilidade recebem o programa, o tratamento é considerado probabilístico.

bys renda: gen bf = cond(v2403==1, peso, 0)

Vamos utilizar o comando ‘preserve’ para trabalhar com a base sem perder as informações originais.

preserve

O comando ‘collapse’, com a opção ‘sum’, soma, na coluna, todos os valores das variáveis à esquerda da vírgula, dentro dos valores idênticos definidos pela variável imediatamente após ‘by’.

collapse (sum) peso bf , by(renda)

g pbf = bf/peso

Vamos observar graficamente se há descontinuidade.

tw lowess pbf renda if renda<=60 || lowess pbf renda if renda>60 ,  xline(60)

tw lowess pbf renda if renda<=120 || lowess pbf renda if renda>120 ,  xline(120)

tw lowess pbf renda if renda<=149 || lowess pbf renda if renda>149 ,  xline(149)

restore

Apenas no terceiro caso temos uma descontinuidade que nos interessa, para efeito de exercício. Daqui pra frente, vamos supor que a elegibilidade é R$ 150,00 per capita.
Vamos salvar a base, para posterior utilização. O comando ‘sort’ é utilizado para ordenar a base de acordo com a identificação do domicílio. Será imprescindível para unir a base de domicílios à de pessoas.

Abra a base ‘nordestepess.dta’.

use data\nordestepess, clear

sum

Vamos fazer uma junção de bases. Para isso, utilizamos o comando ‘merge’. Precisamos sortear a base. O ‘keep’ indica as variáveis que se quer trazer da base ‘nordeste’ para a base ‘nordestepess’.

sort uf v0102 v0103

merge uf v0102 v0103 using data\nordeste, keep(renda treat)

Estamos interessados somente nas observações comuns aos dois arquivos.

tab _merge

keep if _merge==3

drop _merge

Vamos criar nossas variáveis de interesse. 
recode v0602 (2 = 1) (4 = 0), g(escola)

recode v9001 (3 = 0), copyrest g(trabalho)

rename v4729 peso

Novamente, o mesmo procedimento acima, agora com os indivíduos, para verificar a descontinuidade da variável de tratamento.
bys renda: gen bf = cond(treat==1, peso, 0)

preserve

collapse (sum) peso bf , by(renda)

g pbf = bf/peso

tw lowess pbf renda if renda<=60 || lowess pbf renda if renda>60 ,  xline(60)

tw lowess pbf renda if renda<=120 || lowess pbf renda if renda>120 ,  xline(120)

tw lowess pbf renda if renda<=149 || lowess pbf renda if renda>149 ,  xline(149)

restore

Agora, verificando graficamente a descontinuidade das variáveis de interesse.

preserve

collapse (sum) peso escola trabalho , by(renda)

g pescola = escola/peso

g ptrab = trabalho/peso

tw lowess pescola renda if renda<=149 || lowess pescola renda if renda>149 ,  xline(149)

tw lowess ptrab renda if renda<=149 || lowess ptrab renda if renda>149 ,  xline(149)

restore

Podemos testar a descontinuidade da variável de tratamento no ponto de corte.

g h = renda>=148 & renda<=150

reg treat renda if h==1 & renda<=149

loc betal = _b[_cons]

loc sdl = _se[_cons]*e(N)

loc obsl = e(N)

reg treat renda if h==1 & renda>=149

loc betar = _b[_cons]

loc sdr = _se[_cons]*e(N)

loc obsr = e(N)

ttesti `obsr' `betar' `sdr' `obsl' `betal' `sdl'
3.1 Regressão Linear Local 
Vamos direto para a regressão linear local para o caso Fuzzy. Arbitrariamente, estabelecemos um bandwidth de 10.

drop h
g h = renda>=139 & renda<=159

Freqüência a escola.

Passo a Passo

Precisamos de algumas variáveis. Abaixo, criamos o desvio em relação ao ponto de corte.
g xcl = renda - 149 if h==1 & renda<=149

g xcr = renda - 149 if h==1 & renda>149

Regressão para a variável de interesse.

reg escola xcl if xcl~=. [aw=peso]

loc alfayl = _b[_cons]

reg escola xcr if xcr~=. [aw=peso]

loc alfayr = _b[_cons]

Regressão para a variável de tratamento.

reg treat xcl if xcl~=. [aw=peso]

loc alfatl = _b[_cons]

reg treat xcr if xcr~=. [aw=peso]

loc alfatr = _b[_cons]

Calculando o impacto

di (`alfayl' - `alfayr')/ (`alfatl' - `alfatr')

Estimação por MQ2E

Precisamos das variáveis de interação entre renda e se está à direita ou à esquerda do ponto de corte.
g xcli = cond(renda>=139 & renda<=149,renda - 149,0)

g xcri = cond(renda>=150 & renda<=159,renda - 149,0)

O instrumento é se as observações estão do lado esquerdo do ponto de corte.
g instru = renda>=139 & renda<=149

Como vimos, a regressão linear local pode ser realizada utilizando os elegíveis como instrumento, e as variáveis de interação como variáveis exógenas, dentro do intervalo definido pelo bandwidth.

ivregress 2sls escola xcli xcri (treat = instru) if h==1 [aw=peso]

Exercícios

· A aula prática de hoje está no arquivo ‘stata_aula9.do’.

· Com a base votex.dta, repita o exercício incluindo covariáveis e analise.

· Com as bases da pnad, calcule o impacto utilizando o método de regressão não paramétrica na fronteira. Faça ainda a análise, por meio de regressão linear local, do impacto do bolsa-familia no trabalho infantil. Experimente diferentes bandwidths.
