Aula prática n.5 – O uso de algoritmos de emparelhamento: Matching em observáveis
Esta aula prática se inicia com uma revisão da aula 4, uma vez que a base de dados que utilizaremos nesta aula será a mesma.
Assim, antes de iniciarmos o exercício desta aula, novamente vamos descrever uma forma de organizar seus arquivos do Stata. Essa organização será útil tanto para esta quanto para as aulas futuras.

Primeiramente, dentro de seu pen-drive, crie as seguintes pastas:

· “E:\curso\do”

· “E:\curso\log”

· “E:\curso\base”

· “E:\curso\ado’.

Esta última pasta é a grande novidade da aula de hoje, pois é dela que vamos “puxar” arquivos .ado que tenham comandos especiais, como o que utilizaremos hoje para efetuarmos a análise de ATE com base em matching. Em seguida, quem ainda não copiou o arquivo da aula de hoje deve fazer o download do arquivo ‘stata_aula5.do’ no link:

http://www.undp-povertycentre.org/evaluation/praticas/do/
Salve este arquivo em “E:\curso\do”.

O início do arquivo ‘stata_aula5.do’ contém as configurações básicas geralmente utilizadas pelo usuário de Stata:
clear

set mem 100m

cd "E:\curso"

sysdir set PLUS "E:\curso\ado\plus"

sysdir set PERSONAL "E:\curso\ado\personal"

sysdir set OLDPLACE "E:\curso\ado"

1. A base de dados
As bases de dados para esta aula estão no link:

http://www.undp-povertycentre.org/evaluation/praticas/bases/
No caso, faça o download dos seguintes arquivos, lembrando de não salvá-los como .HTML:

· nswre74_treated.dta

· nswre74_control.dta

Salve os arquivos na pasta “E:\curso\base”.
Como anteriormente, a base utilizada para a segunda aula prática do curso refere-se a uma amostra do programa National Supported Work Demonstration (NSW). Este foi um programa de treinamento experimental dirigido a trabalhadores com dificuldades em encontrar uma colocação no mercado de trabalho.

Ao abrir o Stata, através do profile.do, primeiro passo é definir um arquivo log.

log using log\aula5.log, replace

Em seguida, abra o arquivo ‘nswre74_treated.dta’.

use base\nswre74_treated.dta, clear

Propositalmente, apresentamos a base separada em dois arquivos, um contendo a amostra somente de tratados e a outra de controles. Portanto, devemos juntar as duas bases em uma só. Para isso, basta utilizar o comando append.

append using base\nswre74_control.dta
Com um tabulate é possível verificar quantas observações estão no grupo de tratamento e quantas estão no grupo de controle:
tabulate treat

2. Análise descritiva e testes de igualdade de médias
O primeiro passo da aula de hoje é, tal qual como na aula 4, testar se em t=0 as variáveis são semelhantes entre os grupos de casos e controles:
mean age-u75, over(treat)

test ([age]_b[1] = [age]_b[0]) ([edu]_b[1] = [edu]_b[0]) ([black]_b[1] = [black]_b[0]) ([hisp]_b[1] = [hisp]_b[0]) ///

([married]_b[1] = [married]_b[0]) ([re74]_b[1] = [re74]_b[0]) ([re75]_b[1] = [re75]_b[0]) ([u75]_b[1] = [u75]_b[0])

Como vimos na aula passada, uma primeira análise do impacto do programa NSW pode ser dada pela diferença de médias dos rendimentos dos participantes e não-participantes no ano de 1978. Contudo, é importante saber se os grupos de casos e controles partiram de rendimentos semelhantes antes de participarem do programa.

ttest re74, by(treat)

ttest re78, by(treat)

Outra forma, como também vimos na aula 4, é usar métodos de regressão sobre variáveis dummies. O coeficiente associado à dummy de tratamento nos informará o ATE:

regress re78 treat

Sabemos que, a fim de expurgar efeitos fixos aos grupos ou mesmo tendência temporal, uma alternativa interessante para estimação de impacto do tratamento é a técnica de diferença-das-diferenças. Contudo, devemos lembrar que o que está sob teste na técnica DD é diferente do ATE acima estimado: aquela abordagem estima a diferença entre as médias na variável de interesse em período posterior; enquanto a técnica DD mede o diferencial de crescimento (ou diminuição) entre os dois grupos no período posterior. Assim, a hipótese nula do teste em DD não é que não houve crescimento na variável de interesse para o grupo dos tratados, e sim, que este crescimento não foi diferente do verificado no grupo de controle. Obviamente, se ambos os grupos não apresentam diferenças nas variáveis em seus níveis iniciais, DD e estimativa em diferenças dão resultados equivalentes. Como sabemos, podemos criar uma variável referente à diferença de rendimentos entre 1974 e 1978 para, posteriormente, avaliar a DD a partir de regressão ou simples testes de médias: 
gen difre=re78-re74

ttest difre, by(treat)

regress difre treat

Podemos realizar os mesmos procedimentos acima utilizando bootstrap:
bootstrap t=r(t), reps(10): ttest re74, by(treat)

bootstrap t=r(t), reps(10): ttest re78, by(treat)
bootstrap t=r(t), reps(10): difre re78, by(treat)

bootstrap, reps(10): regress re78 treat

bootstrap, reps(10): regress difre treat

3. Seleção em observáveis
Sabemos que o NSW partiu de aleatoriedade, mas suponhamos que não o tivesse, e que os indivíduos se auto-selecionam. Suponhamos ainda, tal qual no artigo de Abadie et al. (2004), que a auto-seleção se dá nos observáveis age, edu, black, hisp, married, re74, re75, u74, e u75. Como vimos na aula, uma possibilidade para estimar o ATE é rodar a regressão com os observáveis e suas interações:
mean age-u75

gen agedmean = (age - 23.37079)*treat

gen edudmean = (edu - 10.19551)*treat

gen blackdmean = (black - .8337079)*treat

gen hispdmean = (hisp - .0876404)*treat

gen marrieddmean = (married - .1685393)*treat

gen re74dmean =  (re74 - 2102.265)*treat

gen re75dmean = (re75 - 1377.138)*treat

gen u74dmean = (u74 - .2674157)*treat

gen u75dmean = (u75 - .505618)*treat  

regress re78 treat age edu black hisp married re74 re75 u74 u75 agedmean edudmean blackdmean hispdmean ///

marrieddmean re74dmean re75dmean u74dmean u75dmean 

Outra possibilidade é o programa “nnmatch”, desenvolvido por Abadie et al. (2004). Este programa permite uma série de opções. Eis a sua sintaxe (para maiores detalhes acerca das opções, consulte o artigo ou o help do Stata):
nnmatch depvar treatvar varlist [weight] [if exp] [in range] 

[, tc({ ate | att | atc })  m(#)  metric(maha|matname)  exact(varlist)

biasadj(bias | varlistadj)  robusth(#) population  level(#)  keep(filename)

replace]

Agora, vamos brincar um pouco com o nnmatch. Vamos estimar primeiro o default do algoritmo:
nnmatch re78 treat age edu black hisp married re74 re75 u74 u75

A partir de agora, estimemos apenas a abordagem DD para este mesmo problema:
nnmatch difre treat age edu black hisp married re74 re75 u74 u75

Agora, vamos trabalhar com as opções do programa. Estimemos um ATT (somente nos tratados), corrigido pelo viés nas variáveis explicativas, casando cada caso com 4 controles, arbitrando a matriz de Mahalanobis para a distância entre as variáveis, e permitindo para heteroscedasticidade entre os grupos de casos e controles.
nnmatch difre treat age edu black hisp married re74 re75 u74 u75, tc(att) ///

m(4) metric(maha) bias(bias) robust(4)

Podemos forçar o mesmo algoritmo para casar indivíduos com o máximo de precisão na variável edu. Isto pode ser feito pelo seguinte comando:
nnmatch difre treat age edu black hisp married re74 re75 u74 u75, tc(att) ///

m(4) metric(maha) bias(bias) robust(4) exact(edu)

Suponha que queiramos estimar os possíveis impactos da expansão do programa para a população hispânica. Vamos, então, estimar um ATC (somente nos controles), corrigido pelo viés nas variáveis explicativas, casando cada caso com 4 controles, arbitrando a matriz de Mahalanobis para a distância entre as variáveis, permitindo para heteroscedasticidade entre os grupos de casos e controles. Contudo, a estimativa agora ocorre somente para o grupo hispânico.
nnmatch difre treat age edu married re74 re75 u74 u75 if hisp==1, tc(atc) ///

m(4) metric(maha) bias(bias) robust(4)

4. O uso de pesos amostrais
Muitas das pesquisas que se usa em avaliação de programas sociais são pesquisas amostrais cujas unidades de observação supostamente representam um número definido de outras unidades de observação. Isto significa que as unidades com maior peso amostral representam mais unidades, e vice versa. O interessante é que o algoritmo de nnmatch também trabalha com os pesos amostrais, mesmo que eles sejam "quebrados". Geremos um peso amostral aleatório para ver como o algoritmo se comporta:
set seed 339487731

generate float weightbroken = 1 + uniform()

Finalmente, estimemos o último procedimento, agora com o emprego de pesos amostrais:
nnmatch difre treat age edu black hisp married re74 re75 u74 u75 [pweight=weightbroken], tc(att) ///

m(4) metric(maha) bias(bias) robust(4) exact(edu)

Exercícios

· Como já mencionado, a aula prática de hoje está no arquivo ‘stata_aula5.do’.

Há uma versão da base de dados NSW que tem um grupo de “candidatos” a grupo de controle retirado do PSID. A base “nswpsid.dta” é uma junção das duas bases de dados que resulta em um conjunto não-aleatório com respeito à distribuição do tratamento. Busque explorar os resultados dos algoritmos acima descritos e outras opções do programa ‘nnmatch’ em um contexto não-aleatório de distribuição do tratamento, e interprete os seus resultados. Dica: há um do-file que contem comandos para a lide com a base NSWPSID, que é o stata_aula5nswpsid.do.’ Salve seus resultados em arquivo do tipo .log. 
