Aula prática n.12 – Revisão

A aula de hoje encontra-se na página abaixo, no arquivo ‘stata_aula12.do’:

http://www.undp-povertycentre.org/evaluation/praticas/do/stata_aula12.do
Como de costume:

clear

set memory 200m

cd "G:\curso"

sysdir set PLUS "G:\ado\plus"

sysdir set PERSONAL "G:\ado\personal"

sysdir set OLDPLACE "G:\ado"

1. Análise Descritiva da base de dados
Vamos abrir um arquivo .log:

log using log\aula12.log

Nesta aula, utilizaremos uma base de dados parecida com a da aula anterior, “pnadne.dta”.
use http://www.undp-povertycentre.org/evaluation/praticas/bases/pnadne.dta, clear

Essa base contém informações para a região Nordeste do Brasil, retiradas da PNAD 2006.
Uma simples descritiva. A variável ‘pbf’ quando assume valor igual a 1 indica que o domicílio recebe benefício do Programa Bolsa Família (ou de programa similar). 
tabulate pbf
tabulate pbf [aw = v4611]

Nesta aula, estaremos interessados em estimar se o programa de transferência de renda afeta a renda domiciliar indiretamente. Ou seja, vamos testar se a participação no Bolsa Família tem impacto sobre a geração autônoma de renda das famílias, o chamado efeito substituição.

Para não contaminar a amostra com um grupo de controle muito distinto daqueles que recebem a Bolsa Família, vamos restringir a amostra de acordo com a própria renda per capita dos domicílios. Antes, vamos verificar onde se encontra este ponto de corte:

histogram netypc if pbf==1 [fw = v4611]
E então restringimos a amostral:

keep if netypc<500
Agora, vamos fazer uma análise descritiva da variável de resultado:
mean netypc [aw = v4611], over(pbf)

E testar a diferença nas médias:
display _b[1] - _b[0]

test _b[1] = _b[0]

Vamos testar também a diferença entre as famílias em relação a outras características:
mean tamdom-uf9 [aw = v4611], over(pbf)

test ([nfam]_b[1] = [nfam]_b[0]) ([tamdom]_b[1] = [tamdom]_b[0]) ([idademin]_b[1] = [idademin]_b[0]) ([idosos]_b[1] = [idosos]_b[0]) ([criancas]_b[1] = [criancas]_b[0]) ([pia]_b[1] = [pia]_b[0]) ([mulher]_b[1] = [mulher]_b[0]) ([idade]_b[1] = [idade]_b[0]) ([idade]_b[1] = [idade]_b[0]) ([branco]_b[1] = [branco]_b[0]) ([preto]_b[1] = [preto]_b[0]) ([anoest]_b[1] = [anoest]_b[0]) ([conjuge]_b[1] = [conjuge]_b[0]) ([densidade]_b[1] = [densidade]_b[0]) ([parede]_b[1] = [parede]_b[0]) ([agua]_b[1] = [agua]_b[0]) ([esgoto]_b[1] = [esgoto]_b[0]) ([lixo]_b[1] = [lixo]_b[0]) ([luz]_b[1] = [luz]_b[0]) ([rural]_b[1] = [rural]_b[0]) ([metrop]_b[1] = [metrop]_b[0]) ([cidade]_b[1] = [cidade]_b[0]) ([uf1]_b[1] = [uf1]_b[0]) ([uf2]_b[1] = [uf2]_b[0]) ([uf3]_b[1] = [uf3]_b[0]) ([uf4]_b[1] = [uf4]_b[0]) ([uf5]_b[1] = [uf5]_b[0]) ([uf6]_b[1] = [uf6]_b[0]) ([uf7]_b[1] = [uf7]_b[0]) ([uf8]_b[1] = [uf8]_b[0]) ([uf9]_b[1] = [uf9]_b[0])
2. Regressão Linear

Antes da análise de regressão, vamos gerar as variáveis de interação:
quietly foreach var of varlist nfam-uf8 {

    summarize `var' [aw = v4611] if e(sample)

    generate M_`var' = pbf*(`var' - r(mean))

}

Assim, podemos estimar o ATE de acordo com o modelo de Rubin (1977), supondo que a seleção para o programa ocorre com base nestas variáveis e a forma paramétrica resolve os problemas de seleção:
regress netypc pbf nfam-uf8 M_* [aw = v4611]
estimates store ATEols
Para estimar o ATT precisamos das variáveis de interação centralizando a média apenas nos tratados.

quietly foreach var of varlist nfam-uf8 {

    summarize `var' [aw = v4611] if pbf==1 & e(sample)

    generate S_`var' = pbf*(`var' - r(mean))

}

regress netypc pbf nfam-uf8 S_* [aw = v4611]

estimates store ATTols
Para verificar o impacto sobre os tratados somente em áreas rurais:
summarize S_rural if e(sample) & rural==1

scalar Rural = r(mean)

lincom _b[pbf] + _b[S_rural]*Rural
3. Uso do Escore de Propensão
Primeiro, vamos calcular o escore de propensão, ou seja, a probabilidade de tratamento condicionado às características do domicílio.
Para selecionar as variáveis do escore de propensão, vamos utilizar o método proposto por Hirano e Imbens (2002). A técnica consiste em pegar aquelas variáveis que tenham uma correlação simples significativa com a variável de tratamento.

global sel = ""

quietly foreach var of varlist nfam-uf9 {


correlate pbf `var' [aw = v4611]

local c`var' = abs(_result(4))

if `c`var''> 0.0105 {



global sel = "$sel `var'"


} 

}

Para estimar a probabilidade, vamos utilizar um modelo probit com as variáveis selecionadas:
logit pbf $sel [pw = v4611]

predict pscore

Agora, definiremos o suporte comum:
summarize pscore if pbf==1

generate comsup = pscore>=r(min) & pscore<=r(max)

summarize pscore if pbf==0

replace comsup = 0 if pscore<r(min) | pscore>r(max)

Vamos conferir quantos domicílios estão fora do suporte comum na amostra:

tabulate comsup

E a renda média destes domicílios:

tabulate comsup [aw = v4611], summarize(netypc)
Calculado o escore de propensão, podemos gerar os pesos para calcular o ATE e o ATT:
generate wate = v4611*((pbf/pscore) + ((1 - pbf)/(1 - pscore))) if comsup==1

generate watt = wate*pscore

Uma forma de verificar o balanço é testando as médias de todas as variáveis de controle ponderadas pelo ‘propensity score’:
mean pscore tamdom-uf9 [aw = wate], over(pbf)

test ([pscore]_b[1] = [pscore]_b[0])

foreach var of varlist tamdom-uf9 {

    display [`var']_b[1] - [`var']_b[0]
    test ([`var']_b[1] = [`var']_b[0])
}

Note que nem todas as variáveis ficaram balanceadas.

Com o propensity score, podemos estimar uma diferença de médias ponderada, obtendo um ATE:

regress netypc pbf [aw = wate]

estimates store ATEw

E um ATT:
regress netypc pbf [aw = watt]

estimates store ATTw

Podemos ainda incluir variáveis de controle na regressão para obtenção do ATE:

regress netypc pbf nfam-uf8 M_* [aw = wate]

estimates store ATEwols
E um ATT:

regress netypc pbf nfam-uf8 S_* [aw = watt]

estimates store ATTwols
Por fim, vamos comparar os resultados:
estimates table ATEols ATEw ATEwols ATTols ATTw ATTwols, star keep(pbf)
4. Regressão Descontínua
Sabendo que o programa beneficia principalmente famílias com crianças menores de 15 anos, podemos considerar isto como um ponto de corte para a elegibilidade.
Claro que não podemos considerar isto como uma descontinuidade sharp, pois é possível que existam domicílios beneficiários sem crianças menores de 15 anos. Mas vamos testar se é possível fazer, pelo menos, uma análise fuzzy.
Primeiro, vamos calcular o bandwidth ótimo. Este cálculo se baseia em um teste de média:

tempvar h t acum r e w acum2 band

generate `h' = abs(idademin - 15.5)

generate `t' = idademin>15

bysort `t' (`h'): g `acum' = sum(v4611)

bysort `t': g `r' = sum(netypc*v4611)/`acum'

generate `e' = (netypc - `r')^2

sort `h'

generate `acum2' = sum(v4611)

generate `w' = sum(`e'*v4611)/`acum2' if `h'!=.

bysort `t' (`h'): replace `w' = . if `h'<3 | `h'>9

sort `w'

generate `band' = `h'[1]
summarize `band'

scalar BW = r(mean)

Agora sabemos qual o bandwidth ótimo, então vamos testar a descontinuidade nos 15 anos de idade, fazendo primeiramente um gráfico:
twoway lpoly pbf idademin if idademin<=15 [aw = v4611] || lpoly pbf idademin if idademin>15 [aw = v4611]
Para testar esta diferença na fronteira:

generate front = idademin<=15 if idademin>=15-BW & idademin<=15+BW
regress pbf front [aw = v4611]
Agora, podemos estimar um fuzzy RD por meio de mínimos quadrados em dois estágios, onde a variável instrumental será a idade da criança mais nova:

generate x1 = (idademin - 15.5)*(1 - front)

generate x2 = (idademin - 15.5)* front

ivreg netypc (pbf = front) x1 x2 nfam idosos pia-uf8 [aw = v4611], robust
estimates store RDD

Agora compare os resultados:
estimates table ATEols ATEw ATEwols ATTols ATTw ATTwols RDD, star

Muito obrigado pela participação, esperamos que as aulas tenham sido de bom proveito. Não esqueçam de fechar o log:
log close

