Aula prática n.4 – Métodos de regressão e diferenças-das-diferenças 

Esta aula prática inicia com uma revisão da aula 2, uma vez que a base de dados que utilizaremos nesta aula será a mesma. 

Assim, antes de iniciar o exercício desta aula, vamos descrever uma forma de organizar seus arquivos do Stata. Essa organização será útil tanto para esta quanto para as aulas futuras.

Primeiramente, dentro de seu pen-drive, crie as seguintes pastas:

· “E:\curso\do”

· “E:\curso\log”

· “E:\curso\base”

Em seguida, quem ainda não o copiou deve fazer o download do arquivo ‘profile.do’ no link:

http://www.undp-povertycentre.org/evaluation/praticas/do/
Salve este arquivo em “E:\curso\do”.

O profile.do contém as configurações básicas geralmente utilizadas pelo usuário de Stata:
clear

set mem 100m

cd "E:\curso"

sysdir set PLUS "E:\ado\plus"

sysdir set PERSONAL "E:\ado\personal"

sysdir set OLDPLACE "E:\ado"

Depois de copiado o arquivo, abra-o no Notepad e modifique o diretório do pen-drive. Salve, feche o arquivo e abra-o novamente, desta vez utilizando o Stata.

1. A base de dados
As bases de dados para esta aula estão no link:

http://www.undp-povertycentre.org/evaluation/praticas/bases/
No caso, faça o download dos seguintes arquivos, lembrando de não salvá-los como .HTML:

· nswre74_treated.dta

· nswre74_control.dta

Salve os arquivos na pasta “E:\curso\base”.
A base utilizada para a segunda aula prática do curso refere-se a uma amostra do programa National Supported Work Demonstration (NSW). Este foi um programa de treinamento experimental dirigido a trabalhadores com dificuldades em encontrar uma colocação no mercado de trabalho.

Ao abrir o Stata, através do profile.do, primeiro passo é definir um arquivo log.

log using log\aula2.log
Em seguida, abra o arquivo ‘nswre74_treated.dta’.

use base\nswre74_treated.dta, clear

Propositalmente, apresentamos a base separada em dois arquivos, um contendo a amostra somente de tratados e a outra de controles. Portanto, devemos juntar as duas bases em uma só. Para isso, basta utilizar o comando append.

append using base\nswre74_control.dta
Com um tabulate é possível verificar quantas observações estão no grupo de tratamento e quantas estão no grupo de controle:
tabulate treat

2. Análise descritiva

Para uma descrição do conteúdo da base de dados, utilize o comando describe:
describe

Para uma análise descritiva geral das variáveis, utilize o comando summarize:
summarize

Além desta descrição geral, podemos fazer uma análise descritiva por sub-amostras. No caso, estamos interessados na estatística descritiva para os grupos de tratamento e controle.

Uma das formas de se fazer isso é: 

sort treat

by treat: summarize 
Outros meios são:

mean age-u75, over(treat)

tabstat age-u75, by(treat) statistics(median mean sd p25) columns(statistics)

3. Teste de igualdade de média
Para testar se a média de determinada variável é diferente entre duas sub-amostras, podemos utilizar o comando ttest:
ttest re78, by(treat)

Outra forma de fazer este teste é:

mean age-u75, over(treat)

test [age]_b[1] = [age]_b[0]
Através dos comandos mean e ttest, podemos também verificar a significância conjunta de diferença nas médias para mais de uma variável:
test ([age]_b[1] = [age]_b[0]) ([black]_b[1] = [black]_b[0]) ([u75]_b[1] = [u75]_b[0]) ([re74]_b[1] = [re74]_b[0])
4. Teste de igualdade de distribuição
Uma maneira de conferir a distribuição de uma variável é fazendo um histograma:
histogram re78, by(treat)

Para testar a igualdade entre duas distribuições de uma mesma variável, é recomendado utilizar o teste Kolmogorov-Smirnov (ksmirnov). Este teste consiste em comparar as diferenças, positivas e negativas, nos valores da variável entre os grupos:
ksmirnov re78, by(treat)
5. Avaliação de Impacto – primeiras impressões

Uma primeira estimativa dos impactos do programa NSW pode ser dada pela abordagem “um grupo antes e depois do tratamento”. Em verdade, como dispomos de um grupo de controle, podemos estimar o suposto impacto do tratamento pela forma de análise contrafactual mais simples: teste de médias ex-post nas variáveis de interesse. Contudo, vale a pena saber se ex-ante ao programa os indivíduos dos grupos de casos e controles tinham as mesmas características. 
ttest re74, by(treat)

ttest re78, by(treat)

Por meio de uma regressão simples (univariada), chegamos ao mesmo resultado (com a diferença em um grau de liberdade):
regress re78 treat

Na aula de hoje aprendemos como estimar o modelo em diferenças-das-diferenças. Uma forma de se estimar o modelo nesta forma é criar uma variável que representa a diferença na variável de interesse (no caso, os rendimentos do trabalho) entre 74 e 78, para procedermos de forma análoga ao anteriormente mostrado:

gen difre=re78-re74

ttest difre, by(treat)

regress difre treat
Obs: A interpretação do modelo DD é distinta do modelo de diferenças simples!
Com respeito à abordagem “um-grupo-antes-e-depois”, uma forma simples de efetuá-la no Stata é estimar a diferença nos rendimentos apenas no grupo dos tratados (uma vez que esta abordagem não utiliza avaliação contrafactual). Na prática, acompanha-se o grupo dos tratados ao longo do tempo e atribui-se a diferença nos rendimentos ao tratamento. Note que o efeito estimado de acordo com este desenho é bem mais alto. De fato, com as estimativas DD, percebemos que a tendência temporal tem um papel muito forte.
sum difre if treat==1

6. Avaliação de Impacto – O uso de bootstrapping

Bootstrapping significa executar um procedimento estatístico diversas vezes com subamostras (com ou sem reposição) com o intuito de estimar parâmetros de interesse de forma consistente. A seguir, estimamos os testes de médias e a regressão a partir de procedimentos de bootstrapping.
bootstrap t=r(t), reps(10): ttest re74, by(treat)

bootstrap t=r(t), reps(10): ttest re78, by(treat)

bootstrap, reps(10): regress re78 treat
7. Diferenças-das-diferenças – O uso de modelos de regressão

As equações que estimam modelos de diferenças-das-diferenças que vimos hoje exigem uma estrutura em dados de painel, e não longitudinais, como é a nossa base de dados original. Assim, precisaremos converter a base para o formato “long”, e não “wide”, na linguagem do Stata. O primeiro passo é criar um identificador para os indivíduos:

gen id=_n
Após este passo, é importante criar o djt do artigo do Meyer. Isto implica em transformar nossa variável de tratamento em uma variável que seja 0 em t0  e assuma valor 1 apenas em t1. Isto pode ser facilmente realizado da seguinte forma:

rename treat treat78

gen treat74=0
Agora, vamos converter nossos dados para a estrutura de painel:

reshape long treat re, i(id) j(year)
Podemos realizar uma avaliação de impacto bem simples, que tira apenas a diferença de médias entre o grupo de casos e controles:

regress re treat
Para estimarmos o modelo de diferenças-das-diferenças, precisamos criar as dummies referentes aos efeitos temporais comuns aos dois grupos e aos efeitos fixos referentes a cada grupo. Como nossa base está ordenada por casos e controles, sabemos que os 260 primeiros indivíduos são controles e o restante são casos. Assim:

gen treatfix=1 if id>260
replace treatfix=0 if id<=260
gen timeeffect=1 if year==78

replace timeeffect=0 if year==74
Agora, podemos finalmente estimar o modelo em diferenças-das-diferenças. Note que o coeficiente estimado é exatamente o mesmo de quando estimamos a equação regress difre treat:
regress re timeeffect treatfix treat 
8. Interações de ordem superior – O uso de modelos de regressão

Suponhamos que um pesquisador esteja interessado em avaliar não exatamente o efeito médio do NSW, mas o impacto do programa sobre grupo composto pelos cidadãos negros. Como vimos, uma forma de se avaliar isto é incorporar interações de ordem superior no modelo de regressão. Para isto, criemos as variáveis, após verificarmos se existe aleatoriedade na distribuição dos cidadãos negros entre os grupos de casos e controles:

ttest black if year==78 , by(treat)

gen blacktime=black*timeeffect 

gen blackfix=black*treatfix 

gen blacktreat=black*treat

Finalmente, estimamos o modelo de regressão. Estamos interessados no coeficiente de blacktreat:

regress re timeeffect treatfix black treat blacktime blackfix blacktreat
Exercícios

· A aula prática de hoje está no arquivo ‘stata_aula4.do’.

· Utilizando a base ‘nls.dta’, disponível em

http://www.undp-povertycentre.org/evaluation/praticas/bases/,

repita o exercício acima, verificando se é possível tratar este caso como um experimento. O interesse é avaliar o impacto da sindicalização (variável ‘union’) sobre salário (variável ‘ln_wage’).

Dica: estime o impacto para anos específicos.
