Aula prática n.11 – Revisão 
A aula de hoje encontra-se na página abaixo, no arquivo ‘stata_aula11’:
 http://www.undp-povertycentre.org/evaluation/praticas/do/. 

Como de costume:

clear

set mem 300m

cd "E:\curso"

sysdir set PLUS "E:\ado\plus"

sysdir set PERSONAL "E:\ado\personal"

sysdir set OLDPLACE "E:\ado"

1. A base de dados
Nesta aula, utilizaremos a base “criancas.dta”.
http://www.undp-povertycentre.org/evaluation/praticas/bases/
Faça o download, lembrando de NÃO salvá-los como .HTML.
Essa base contém informações para a região nordeste do Brasil, retiradas da PNAD 2006. Inclui os domicílios com renda per capita até R$ 230,00, traz dados sobre crianças entre 7 e 15 anos, além de características da pessoa de referência e das condições do domicílio.
Vamos abrir um arquivo .log.
log using log\aula11.log

use base\criancas.dta, clear

Uma simples descritiva. A variável ‘tratado’ quando assume valor igual a 1 indica que o a criança pertence a um domicílio beneficiário do programa bolsa-família. 
tab tratado 

tab tratado [aw=peso]

tab trabalha tratado [aw=peso], cell nofreq

Abaixo temos a proporção de crianças que trabalham para tratados e controles. Temos também um teste de média.
mean trabalha [aw=peso], over(tratado)

di _b[1] - _b[0]

test _b[1] = _b[0]

2. Regressão Linear

Lembrando que a diferença entre essas médias pode ser obtida por regressão.
regress trabalha tratado [aw=peso]

estimates store OLS

Vamos incluir alguns controles referentes às características das crianças. Antes, façamos um teste de média dessas variáveis. 
mean sexo-anoest2  [aw=peso], over(tratado)

test ([sexo]_b[1] = [sexo]_b[0]) ([raca]_b[1] = [raca]_b[0]) ([idade]_b[1] = [idade]_b[0]) ([idade2]_b[1] = [idade2]_b[0]) ([anoest]_b[1] = [anoest]_b[0]) ([anoest2]_b[1] = [anoest2]_b[0])

regress trabalha tratado sexo raca idade idade2 anoest anoest2 [aw=peso]

Agora, incluímos controles referentes às características da pessoa de referencia e do domicílio.

regress trabalha tratado sexo raca idade idade2 anoest anoest2 sexop racap idadep idade2p anoestp anoest2p comcart semcart trabdom cp ncri5 ncri15 renda [aw=peso]

estimates store OLSM

Podemos incluir também variáveis de interação, uma vez que apenas aqueles controles podem não ser suficiente. 

foreach var of varlist sexo raca idade idade2 anoest anoest2 sexop racap idadep idade2p anoestp anoest2p comcart semcart trabdom cp ncri5 ncri15 renda {

    summarize `var' [aw=peso] if e(sample)

    generate M_`var' = tratado*(`var' - r(mean))

}

Chegamos ao ATE.

regress trabalha tratado sexo raca idade idade2 anoest anoest2 sexop racap idadep idade2p anoestp anoest2p comcart semcart trabdom cp ncri5 ncri15 renda M* [aw=peso]

estimates store ATE

Para estimar o ATT precisamos das variáveis de interação centralizando a média apenas nos tratados.

foreach var of varlist sexo raca idade idade2 anoest anoest2 sexop racap idadep idade2p anoestp anoest2p comcart semcart trabdom cp ncri5 ncri15 renda {

    summarize `var' [aw=peso] if tratado==1 & e(sample)

    generate S_`var' = tratado*(`var' - r(mean))

}

regress trabalha tratado sexo raca idade idade2 anoest anoest2 sexop racap idadep idade2p anoestp anoest2p comcart semcart trabdom cp ncri5 ncri15 renda S* [aw=peso]

estimates store ATT

A heterogeneidade do impacto pode ser obtida da seguinte forma. 

summarize M_sexo if e(sample) & sexo==1

scalar Sexo = r(mean)

estimates restore ATE

lincom _b[tratado] + _b[M_sexo]*Sexo

3. Matching e Escore de Propensão e Regressão Ponderada (IPW)
Podemos estimar o impacto por ‘matching’. O comando abaixo utiliza a abordagem do vizinho mais próximo, ou seja, faz o pareamento de tratados e controles de acordo com as variáveis especificadas, com o número de vizinhos escolhido. (Para o lar!!)

nnmatch trabalha tratado sexo idade anoest anoest2 [pweight=v4729], tc(att) m(4) metric(maha) bias(bias) robust(4)

Agora, vamos estimar o impacto utilizando o ‘propensity score matching’. Primeiro, precisamos estimar a probabilidade de tratamento. Vamos selecionar as variáveis de acordo com sua correlação com a variável de tratamento.
global sel = ""

foreach var of varlist ncri5 ncri15 renda idadep anoestp sexop idade2p racap anoest2p comcart semcart trabdom cp trabagri lnmorador parede dorm2 dorm3 moradia aguacanal banheiro redecolet fossarud lixo1 lixo2 luz fonefixo fogao combcoz filtro geladeira area {


corr tratado `var' [aw=peso]


loc c`var' = abs(_result(4))


if `c`var''>.05 {



global sel = "$sel `var'"


} 

}

Estimando a probabilidade, utilizando as variáveis significantes na correlação.

logit tratado $sel [pw=peso]

predict pscore

Construímos o suporte comum.

sum pscore if tratado==1

gen comsup = pscore>=r(min) & pscore<=r(max)

sum pscore if tratado==0

replace comsup = 0 if pscore<r(min) | pscore>r(max)

E geramos os pesos para calcular o ATE e o ATT.
gen wpeso = (tratado/pscore) + ((1 - tratado)/(1 - pscore)) if comsup==1

gen wate = wpeso*peso

generate watt = wate*pscore

Uma forma de verificar o balanço é testando as médias ponderadas pelo ‘propensity score’.

mean pscore $sel [aw=wate], over(tratado)

test ([pscore]_b[1] = [pscore]_b[0])

test ([ncri5]_b[1] = [ncri5]_b[0]) ([ncri15]_b[1] = [ncri15]_b[0]) ([renda]_b[1] = [renda]_b[0]) ([sexop]_b[1] = [sexop]_b[0]) ([anoestp]_b[1] = [anoestp]_b[0]) ([anoest2p]_b[1] = [anoest2p]_b[0]) 

([comcart]_b[1] = [comcart]_b[0]) ([cp]_b[1] = [cp]_b[0]) ([trabagri]_b[1] = [trabagri]_b[0]) ([lnmorador]_b[1] = [lnmorador]_b[0]) ([dorm3]_b[1] = [dorm3]_b[0]) ([aguacanal]_b[1] = [aguacanal]_b[0]) ([banheiro]_b[1] = [banheiro]_b[0]) ([redecolet]_b[1] = [redecolet]_b[0]) ([lixo1]_b[1] = [lixo1]_b[0]) ([lixo2]_b[1] = [lixo2]_b[0]) ([luz]_b[1] = [luz]_b[0]) ([fonefixo]_b[1] = [fonefixo]_b[0]) ([fogao]_b[1] = [fogao]_b[0]) ([combcoz]_b[1] = [combcoz]_b[0]) ([geladeira]_b[1] = [geladeira]_b[0]) ([area]_b[1] = [area]_b[0])

Vamos estimar o ATT utilizando a abordagem do vizinho mais próximo.

watt trabalha tratado [aw=peso], pscore(pscore) comsup

Agora, estimamos o ATT utilizando a abordagem semiparamétrica, o IPW (Inverse Probability Weighting). Precisamos criar novamente as variáveis de interação, agora considerando o suporte comum.

reg trabalha tratado [aw=watt]

foreach var of varlist sexo raca idade idade2 anoest anoest2 sexop racap idadep idade2p anoestp anoest2p comcart semcart trabdom cp ncri5 ncri15 renda {

    summarize `var' [aw=peso] if e(sample) & tratado==1
    generate SS_`var' = tratado*(`var' - r(mean))

}

regress trabalha tratado sexo raca idade idade2 anoest anoest2 sexop racap idadep idade2p anoestp anoest2p comcart semcart trabdom cp ncri5 ncri15 renda SS* [aw=watt]

estimates store WATT

O ATE é calculado da mesma forma, utilizando o peso apropriado.

foreach var of varlist sexo raca idade idade2 anoest anoest2 sexop racap idadep idade2p anoestp anoest2p comcart semcart trabdom cp ncri5 ncri15 renda {

    summarize `var' [aw=peso] if e(sample)

    generate MM_`var' = tratado*(`var' - r(mean))

}

regress trabalha tratado sexo raca idade idade2 anoest anoest2 sexop racap idadep idade2p anoestp anoest2p comcart semcart trabdom cp ncri5 ncri15 renda MM* [aw=wate]

estimates store WATE

estimates table OLS OLSM ATE ATT WATT WATE, star

