Aula prática n.6 – Propensity Score Matching em observáveis
Esta aula seguirá a problemática da aula 5, que trabalhou os algoritmos de hard matching em observáveis. 
Assim, antes de iniciarmos o exercício desta aula, novamente vamos descrever uma forma de organizar seus arquivos do Stata. Essa organização será útil tanto para esta quanto para as aulas futuras.

Primeiramente, dentro de seu pen-drive, crie as seguintes pastas:

· “E:\curso\do”

· “E:\curso\log”

· “E:\curso\base”

· “E:\curso\ado’.

Esta última pasta é a grande novidade da aula de hoje, pois é dela que vamos “puxar” arquivos .ado que tenham comandos especiais, como o que utilizaremos hoje para efetuarmos a análise de ATE com base em matching. Atentem para os arquivos “pscore.ado” e “att*.ado”. Em seguida, quem ainda não copiou o arquivo da aula de hoje deve fazer o download do arquivo ‘stata_aula6.do’ no link:

http://www.undp-povertycentre.org/evaluation/praticas/do/
Salve este arquivo em “E:\curso\do”.

O início do arquivo ‘stata_aula6.do’ contém as configurações básicas geralmente utilizadas pelo usuário de Stata:
clear

set mem 450m

cd "E:\curso"

sysdir set PLUS "E:\curso\ado\plus"

sysdir set PERSONAL "E:\curso\ado\personal"

sysdir set OLDPLACE "E:\curso\ado"
1. A base de dados
As bases de dados para esta aula estão no link:

http://www.undp-povertycentre.org/evaluation/praticas/bases/
No caso, faça o download do seguinte arquivo, lembrando de não salvá-los como .HTML:

· Nswpsid.dta
Salve os arquivos na pasta “E:\curso\base”.
Como anteriormente, a base utilizada para esta aula prática do curso refere-se a uma amostra do programa National Supported Work Demonstration (NSW) em conjunto com a base do PSID. Sairão da base do PSID os candidatos ao grupo de controle do NSW, e a base final não representa mais um experimento aleatório.
Definamos o arquivo de  log e puxemos a base de dados que vamos utilizar:
log using log\aula6.log, replace

use base\nswpsid.dta, clear
Com um tabulate é possível verificar quantas observações estão no grupo de tratamento e quantas estão no grupo de controle:
tabulate TREAT
2. Análise descritiva e testes de igualdade de médias
Já sabemos o resultado, mas vamos começar testando se as variáveis em t0 entre os grupos de casos e controles são iguais:
mean AGE-U75, over(TREAT)

test ([AGE]_b[1] = [AGE]_b[0]) ([EDUC]_b[1] = [EDUC]_b[0]) ([BLACK]_b[1] = [BLACK]_b[0]) ([HISP]_b[1] = [HISP]_b[0]) ///

([MARR]_b[1] = [MARR]_b[0]) ([RE74]_b[1] = [RE74]_b[0]) ([RE75]_b[1] = [RE75]_b[0]) ([U75]_b[1] = [U75]_b[0])
Como vimos na aula passada, uma primeira análise do impacto do programa NSW pode ser dada pela diferença de médias dos rendimentos dos participantes e não-participantes no ano de 1978. Contudo, é importante saber se os grupos de casos e controles partiram de rendimentos semelhantes antes de participarem do programa.

ttest RE74, by(TREAT)

ttest RE78, by(TREAT)

Outra forma, como também vimos na aula 4, é usar métodos de regressão sobre variáveis dummies. O coeficiente associado à dummy de tratamento nos informará o ATE:

regress RE78 TREAT
Sabemos que, a fim de expurgar efeitos fixos aos grupos ou mesmo tendência temporal, uma alternativa interessante para estimação de impacto do tratamento é a técnica de diferença-das-diferenças. Contudo, devemos lembrar que o que está sob teste na técnica DD é diferente do ATE acima estimado: aquela abordagem estima a diferença entre as médias na variável de interesse em período posterior; enquanto a técnica DD mede o diferencial de crescimento (ou diminuição) entre os dois grupos no período posterior. Assim, a hipótese nula do teste em DD não é que não houve crescimento na variável de interesse para o grupo dos tratados, e sim, que este crescimento não foi diferente do verificado no grupo de controle. Obviamente, se ambos os grupos não apresentam diferenças nas variáveis em seus níveis iniciais, DD e estimativa em diferenças dão resultados equivalentes. Como sabemos, podemos criar uma variável referente à diferença de rendimentos entre 1974 e 1978 para, posteriormente, avaliar a DD a partir de regressão ou simples testes de médias: 
gen difre=RE78-RE74

ttest difre, by(TREAT)

regress difre TREAT
Podemos realizar os mesmos procedimentos acima utilizando bootstrap:
bootstrap t=r(t), reps(10): ttest RE74, by(TREAT)

bootstrap t=r(t), reps(10): ttest RE78, by(TREAT)

bootstrap, reps(10): regress RE78 TREAT
bootstrap t=r(t), reps(10): ttest difre, by(TREAT)

bootstrap, reps(10): regress difre TREAT

3. Seleção em observáveis
Sabemos que o NSW o grupo caso-controle partiu de aleatoriedade, mas ao juntarmos a base com o PSID esta aleatoriadade se perde. Contudo, suponhamos que os indíviduos se auto-selecionem. Suponhamos ainda, tal qual no artigo de Abadie et al. (2004), que a auto-seleção se dá nos observáveis AGE, EDUC, BLACK, HISP, MARR, RE74, RE75, U74, e U75. 

Como vimos na aula, uma possibilidade para estimar o ATE é rodar a regressão com os observáveis e suas interações. Como vimos na aula passada, uma possibilidade para estimar o ATE é rodar a regressão com os observáveis e suas interações:
mean AGE-U75

gen agedmean = (AGE -  34.22579)*TREAT

gen edudmean = (EDUC - 11.99439)*TREAT

gen blackdmean = (BLACK -  .2915888)*TREAT

gen hispdmean = (HISP - .0343925)*TREAT

gen marrieddmean = (MARR - .8194393)*TREAT

gen re74dmean =  (RE74 - 18230)*TREAT

gen re75dmean = (RE75 - 17850.89)*TREAT

gen u74dmean = (U74 -   .1293458)*TREAT

gen u75dmean = (U75 -  .1345794)*TREAT  

regress RE78 TREAT AGE EDUC BLACK HISP MARR RE74 RE75 U74 U75 agedmean edudmean blackdmean hispdmean ///

marrieddmean re74dmean re75dmean u74dmean u75dmean 

regress difre TREAT AGE EDUC BLACK HISP MARR U74 U75 agedmean edudmean blackdmean hispdmean ///

marrieddmean u74dmean u75dmean, robust
*Detalhe: a estimativa DD não devem levar em conta os níveis iniciais da variável de interesse, caso contrário, estimaremos o mesmo modelo em nível.

Vimos na aula passada que outra possibilidade é o programa nnmatch, desenvolvido por Abadie et al. (2004). Estimemos um ATT para DD com hardmatch, nas variáveis acima descritas, com a opção para heteroscedasticidade, distância de Mahalanobis e correção de viés:
nnmatch difre TREAT AGE EDUC BLACK HISP MARR RE74 RE75 U74 U75, tc(att) ///

m(4) metric(maha) bias(bias) robust(4)
4. O uso de PS e PSM
Agora vamos estimar o ATT pelos algoritmos de Becker e Ichino. A nossa base de dados já tem as variáveis quadráticas e algumas interações que serão necessárias para replicarmos os modelos do artigo. A sintaxe do programa “pscore.ado” (um programa para estimarmos o propensity score) é:
pscore treatment varlist [weight] [if exp] [in range], pscore(newvar) 

blockid(newvar) detail logit comsup level(#) numblo(#)
Testando o algoritmo do Stata Journal, sem pesos amostrais:

pscore TREAT AGE AGESQ EDUC EDUCSQ MARR BLACK HISP RE74 RE75 RE74SQ RE75SQ U74BLACK, ///

pscore(myscore) blockid(myblock) /*detail*/ comsup numblo(5) level (0.005) logit

Os argumentos "pscore" e "blockid" são compulsórios, uma vez que eles criarão as variáveis da probabilidade estimada e do identificador de bloco. O identificador de bloco são as faixas de propensity score sobre as quais os testes de média ocorrerão. Lembre-se que, ao aumentar o número de blocos (numblo) têm-se maior chance de se atingir as condições de base - basement conditions. A opção detail pode ser interessante se quisermos ver os testes de média dentro dos blocos 1 a 1, mas ele já indica em quais variáveis esta condição é violada. O level indica o rigor dos testes de hipótese de que as variáveis do probit ou logit são iguais dentro dos blocos. Estimemos agora os efeitos de mudanças marginais:
Mfx
Podemos estimar um modelo kitchen-sink usando o propensity score. Contudo, primeiro precisamos criar a variável de interação:

sum myscore

gen myscoredmean = (myscore -  .0691589)*TREAT

sum myscoredmean
regress RE78 TREAT myscore myscoredmean, robust

regress difre TREAT myscore myscoredmean, robust
Finalmente, vamos começar a estimar os ATT com base em PSM. A sintaxe básica do comando de radius matching é dada por:

attr outcome treatment varlist [weight] [if exp] [in range] ,

pscore(scorevar) logit index radius(#) comsup detail bootstrap

reps(#) noisily dots
Comecemos então com um radius matching com distâncias de probabilidade não superiores a 0.01, suporte comum, bootstrap com 10 reposições:

attr difre TREAT AGE AGESQ EDUC EDUCSQ MARR BLACK HISP RE74 RE75 RE74SQ RE75SQ U74BLACK,  ///

pscore(myscore) comsup boot reps(10) dots logit radius(0.01)
Vimos que a literatura aponta que o kernel matching estima a variância de forma mais robusta que o radius matching. A sintaxe do kernel matching é a seguinte:
attr outcome treatment varlist [weight] [if exp] [in range] ,

pscore(scorevar) logit index epan bwidth(#) comsup detail bootstrap

reps(#) noisily dots
A única diferença de sintaxe em relação ao radius matching são as opções “epan” e “bwidth”, que se referem à escolha do Kernel de Epanechnikov ao invés do gaussiano, que é o default. A opção “bwidth” só faz sentido com a opção “epan”, que vai determinar a largura da banda deste Kernel. O default é 0,06. Estimemos pois o kernel matching, usando o kernel epan:
attk difre TREAT AGE AGESQ EDUC EDUCSQ MARR BLACK HISP RE74 RE75 RE74SQ RE75SQ U74BLACK,  ///

pscore(myscore) comsup epan boot reps(10) dots logit

5. O uso dos pesos amostrais

Como estes programas lidam com os pesos amostrais? Novamente, criemos pesos aleatórios:
set seed 339487731

generate float weightbroken = 1 + uniform()

attk difre TREAT AGE AGESQ EDUC EDUCSQ MARR BLACK HISP RE74 RE75 RE74SQ RE75SQ U74BLACK [pweight=weightbroken],  ///

pscore(myscore) comsup boot reps(10) dots logit

Preste atenção à mensagem de erro do programa - de fato, no bootstrapping os pesos amostrais podem perder o sentido.

Como ele não estima as variâncias do kernel matching analiticamente na presença de pesos amostrais, a opção é fazer o seguinte, contando que o pscore já levou em conta os pesos amostrais:
pscore TREAT AGE AGESQ EDUC EDUCSQ MARR BLACK HISP RE74 RE75 RE74SQ RE75SQ U74BLACK [pweight=weightbroken], ///

pscore(myscore) blockid(myblock) /*detail*/ comsup numblo(5) level (0.005) logit

attk difre TREAT,  ///

pscore(myscore) comsup epan boot reps(10) dots
No caso do radius matching, podemos fazer o seguinte:

attr difre TREAT AGE AGESQ EDUC EDUCSQ MARR BLACK HISP RE74 RE75 RE74SQ RE75SQ U74BLACK [pweight=weightbroken],  ///

pscore(myscore) comsup logit radius(0.01)

Veja que as opções boot, reps() e dots foram removidas.
Exercícios

· Como já mencionado, a aula prática de hoje está no arquivo ‘stata_aula6.do’.

· Explore diferentes opções dos programas “att*.ado” na mesma base de dados, e compare seus resultados.
